A Polya Random Walk On A Lattice

Finding The Probability Of Ever Reaching A Specified Lattice Point

John Snyder
January 31, 2015

Problem

A random walk on the 2-dimensional integer lattice begins at the origin. At each step, the walker moves one unit either left, right, or up, each with probability 1/3; no downward steps are ever allowed. A walk is a success if it reaches the point {1, 1}. What is the probability of success?

Note: One can vary the problem by varying the target point; e.g., use {1, 0} or {0, 1} instead. Perhaps there is a good method to resolve the general case of target $\{x, y\}$?

Solution

Summary Of Results

Although the problem can be solved using elementary methods, we will employ the powerful technique of lattice Green functions to solve this problem of a Polya random walk on an integer lattice. This method is developed in great detail in the two volumes by Hughes (see References).

Using lattice Green functions we find that the probability of reaching the lattice point {1, 1} is 2/5. As an illustration of how this method of solution can be generalized to any lattice point, we also compute that the probability of reaching the point {3, 5} is exactly 532/3125. In another example we compute the probability of ever reaching the point {12, 12} to be 2 704 156/244 140 625. We find that the probability of reaching the point {7, 11} is 3 012 672/48 828 125. Finally, the probability of ever reaching the point {11, 7} is irrational, the solution being:

$$1512\sqrt{5} - \frac{264\,135\,048}{78\,125} \approx 0.00616758$$

All of these results have been validated by performing simulation experiments.

In the case where $x = n$ and $y = 0$ the probability that the random walker will ever reach the point $\{n, 0\}$ is given by the following simple expression where ϕ is the golden ratio:

$$1/\phi^{2n}$$

In the case where $x = n$ and $y = 1$ the probability that the random walker will ever reach the point $\{n, 1\}$ is given by the following expression:

$$\frac{1}{5}\left(\frac{2}{3+\sqrt{5}}\right)^n \left(\sqrt{5}\, n + 3\right)$$

In the case where $x = y = n$ the probability that the random walker will ever reach the point $\{n, n\}$ is given by the following simple expression:

$$5^{-n} \binom{2n}{n}$$

Finally, the general formula giving the probability that the random walker will ever reach the lattice point $\{x, y\}$ is given by the following expression in terms of the hypergeometric ${}_2F_1$ regularized function when $\xi \to 1$.

$$\frac{\sqrt{9-4\,\xi^2}}{3^{|x|+y+1}\,y!}\,(|x|+y)!\,\xi^{|x|+y}\,{}_2\tilde{F}_1\left(\frac{1}{2}(y+|x|+1),\,\frac{1}{2}(y+|x|+2);\,|x|+1;\,\frac{4\,\xi^2}{9}\right)$$

Note this general formula arises from the fact that the probability generating function giving the probability that a random walker will *first* arrive at the lattice point $\{x, y\}$ can be obtained from the following integral for $x \geq 0$:

$$\frac{\sqrt{9-4\,\xi^2}}{3^{y+1}\,y!}\int_0^\infty e^{-t}\,(\xi\,t)^y\,I_x\left(\frac{2\,t\,\xi}{3}\right)dt$$

Here I_x is the modified Bessel function.

Theoretical Overview

Hughes develops the detailed theory in his book; here we give only a very brief sketch of his methods.

The walker moves on an infinite d dimensional lattice and we specify his position in terms of the vector $\mathbf{1}$. The probability that the walker is at position $\mathbf{1}$ after n steps is $P_n(\mathbf{1})$. Consistent with this notation we define the probability $p(\mathbf{1})$ that a given step results in a vector displacement of $\mathbf{1}$. The recurrence relation describing the evolution of the walk is:

$$P_{n+1}(\mathbf{1}) = \sum_{\mathbf{1'}} p(\mathbf{1} - \mathbf{1'})\,P_n(\mathbf{1'}) \tag{1}$$

To solve this equation we introduce the discrete Fourier transform and write

$$\tilde{P}_n(\boldsymbol{k}) = \sum_{\mathbf{1}} \exp(i\,\mathbf{1}\cdot\boldsymbol{k})\,P_n(\mathbf{1}) \tag{2}$$

We define the *structure function* of the walk to be

$$\lambda(\boldsymbol{k}) = \sum_{\mathbf{1}} \exp(i\,\mathbf{1}\cdot\boldsymbol{k})\,p(\mathbf{1}) \tag{3}$$

The discrete Fourier transform of (1) is

$$\tilde{P}_{n+1}(\boldsymbol{k}) = \lambda(\boldsymbol{k})\,\tilde{P}_n(\boldsymbol{k}) \tag{4}$$

with the initial condition that we start at the origin we then have $\tilde{P}_0(\boldsymbol{k}) = 1$ so it follows that

$$\tilde{P}_n(\boldsymbol{k}) = \lambda\,(\boldsymbol{k})^n \tag{5}$$

The discrete Fourier transform can be inverted so that

$$P_n(\mathbf{1}) = \frac{1}{(2\pi)^d} \int\int \ldots \int_B \exp(-i\,\mathbf{1}\cdot\boldsymbol{k})\,\tilde{P}_n(\boldsymbol{k})\,d^d\boldsymbol{k} \tag{6}$$

where d is the number of dimensions and B is the first Brillouin zone ($B = [-\pi, \pi]^d$). Now making the substitution (5) into (6) we arrive at the formal solution to the random walk problem

$$P_n(\mathbf{1}) = \frac{1}{(2\pi)^d} \int\int \ldots \int_B \exp(-i\,\mathbf{1}\cdot\boldsymbol{k})\,\lambda(\boldsymbol{k})^n\,d^d\boldsymbol{k} \tag{7}$$

Now the lattice Green generating function (similar to that encountered in the solution of differential equations involving potentials) $P(\mathbf{1}, \xi)$ may be written as

$$P(\mathbf{1}, \xi) = \sum_{n=0}^{\infty} P_n(\mathbf{1})\,\xi^n \tag{8}$$

Substituting (7) into (8), interchanging the orders of integration and summation (the series being absolutely convergent), and summing the resulting geometric series we arrive at the key equation

$$P(\mathbf{1}, \xi) = \frac{1}{(2\pi)^d} \int\int \ldots \int_B \frac{\exp(-i\,\mathbf{1}\cdot\boldsymbol{k})\,d^d\boldsymbol{k}}{1 - \xi\,\lambda(\boldsymbol{k})} \tag{9}$$

The probability of being at the origin after any number of steps can then be found by taking the limit $\xi \to 1$ from below and replacing the vector **1** with $\mathbf{0} = \{0, 0, 0\}$ to obtain

$$P(\mathbf{0}, 1^-) = \frac{1}{(2\pi)^d} \int\int \ldots \int_B \frac{d^d\boldsymbol{k}}{1 - \lambda(\boldsymbol{k})} \tag{10}$$

In a similar manner the probability of being at other locations on the lattice can also be found.

The Probability Of Reaching {1, 1}

Analytic Solution

We first calculate the structure function $\lambda(\boldsymbol{k})$, where $\boldsymbol{k}$ is the two dimensional vector with components $\{x, y\}$. The possible steps are very simple.

```
stepsSC = {{-1, 0}, {1, 0}, {0, 1}};
```

So for the specific two dimensional lattice of this problem the structure function becomes:

```
1/Length[stepsSC] Plus @@ Map[Exp, {ⅈ x, ⅈ y}.# & /@ stepsSC] // ExpToTrig // FullSimplify
```

$$\frac{1}{3}\left(e^{i y} + 2\,\mathrm{Cos}[x]\right)$$

The lattice Green function which gives the probability generating function that a random walker will be found at the vector position **1** after any number of steps as developed in Hughes (3.123, p. 138) is:

$$P(\mathbf{1}, \xi) = \sum_{n=0}^{\infty} P_n(\mathbf{1})\,\xi^n = \frac{1}{(2\pi)^d} \int\int \cdots \int_B \frac{\exp(-i\,\mathbf{1}\cdot\boldsymbol{k})\,d^d\boldsymbol{k}}{1 - \xi\,\lambda(\boldsymbol{k})} \tag{11}$$

where the integrals are taken over the first Brillouin zone of $[-\pi, \pi]^d$ and d is the dimension of the problem.

Computing the Green function at the origin where $\{x, y\} = \{0, 0\}$ we can integrate symbolically to obtain.

```
Assuming[0 < ξ < 1,
 3/(4 π^2) Integrate[1/(3 - ξ (ⅇ^(ⅈ y) + 2 Cos[x])), {x, -π, π}, {y, -π, π}]]
```

$$\frac{3}{\sqrt{9 - 4\,\xi^2}}$$

This same result can be expressed as a single dimensional integral. First notice that the $1/(1 - \xi\,\lambda(\boldsymbol{k}))$ of the integrand in the double integral of formula (11) can be written as the following integral in the case of a random walker on the lattice:

```
Assuming[0 < ξ < 1,
 FullSimplify@Integrate[Exp[-(1 - ξ/3 (ⅇ^(ⅈ y) + 2 Cos[x])) t], {t, 0, ∞}]]
```

$$\text{ConditionalExpression}\left[-\frac{3}{-3 + e^{i y}\,\xi + 2\,\xi\,\text{Cos}[x]},\ \xi\,\text{Re}\left[e^{i y} + 2\,\text{Cos}[x]\right] < 3\right]$$

Plugging this into (11) and integrating out the $\{x, y\}$ we have the following at the origin when $\{x, y\} = \{0, 0\}$.

```
Assuming[0 < ξ < 1 && t > 0,
 HoldForm[Integrate[##, {t, 0, ∞}]] &@@
  {1/(2 π)^2 Integrate[Exp[-(1 - ξ/3 (ⅇ^(ⅈ y) + 2 Cos[x])) t] , {x, -π, π}, {y, -π, π} ]}]
```

$$\int_0^\infty e^{-t}\,\text{BesselI}\left[0, \frac{2\,t\,\xi}{3}\right] dt$$

Note that this single dimensional integral involves the modified Bessel function `BesselI`. Integrating this expression we obtain a generating function for the probability that the walker will be found at the origin after a specified number of steps.

```
Assuming[0 < ξ < 1,
 Integrate[ⅇ^-t BesselI[0, (2 t ξ)/3], {t, 0, ∞}]]
```

$$\frac{3}{\sqrt{9 - 4\,\xi^2}}$$

Note that this result is exactly the same as that found by integrating the original two dimensional integral.

Now at the lattice point $\{1, 1\}$ we have the following similar expression:

```
Assuming[0 < ξ < 1 && t > 0,
  HoldForm[Integrate[##, {t, 0, ∞}]] & @@ {1/(2 π)^2
      Integrate[Exp[-(1 - ξ/3 (ⅇ^(ⅈ y) + 2 Cos[x])) t] Exp[-ⅈ (x + y)], {x, -π, π}, {y, -π, π}]}]
```

$$\int_0^\infty \frac{1}{3} e^{-t}\, t\, \xi\, \mathrm{BesselI}\left[1, \frac{2 t \xi}{3}\right] dt$$

Integrating this expression we obtain a generating function for the probability that the walker will be found at the point {1, 1} after a specified number of steps.

```
Assuming[0 < ξ < 1,
  Integrate[1/3 ⅇ^-t t ξ BesselI[1, 2 t ξ/3], {t, 0, ∞}]]
```

$$\frac{6 \xi^2}{\left(9 - 4 \xi^2\right)^{3/2}}$$

To obtain a generating function for the probability that the random walker will *first* arrive at the lattice point {1, 1} after a specified number of steps we simply take the ratio of the these two generating functions.

```
(6 ξ^2/(9 - 4 ξ^2)^(3/2)) / (3/Sqrt[9 - 4 ξ^2]) // Simplify
```

$$\frac{2 \xi^2}{9 - 4 \xi^2}$$

If we evaluate this generating function at $\xi \to 1$ we obtain the probability that the random walker will ever reach the lattice point {1, 1}, this being the solution to the originally posed problem.

```
2 ξ^2/(9 - 4 ξ^2) /. ξ → 1
```

$$\frac{2}{5}$$

Expanding this generating function as a Taylor's series about the point $\xi = 0$ we can read off the exact probabilities that the random walker *first* reaches the point {1, 1} after the indicated number of steps.

```
Series[2 ξ^2/(9 - 4 ξ^2), {ξ, 0, 18}]
```

$$\frac{2 \xi^2}{9} + \frac{8 \xi^4}{81} + \frac{32 \xi^6}{729} + \frac{128 \xi^8}{6561} + \frac{512 \xi^{10}}{59049} + \frac{2048 \xi^{12}}{531441} + \frac{8192 \xi^{14}}{4782969} + \frac{32768 \xi^{16}}{43046721} + \frac{131072 \xi^{18}}{387420489} + O[\xi]^{19}$$

As an example, we see from this expression that the probability that a random walker *first* reaches the lattice point {1, 1} after exactly 10 steps is $512/59049$.

By taking the derivative of this generating function, evaluating it at $\xi \to 1$, and multiplying by the normalizing constant we obtain the expected number of steps required for the random walker to reach the point {1, 1}, given that he ever reaches this point.

```
5/2 D[2 ξ^2/(9 - 4 ξ^2), ξ] /. ξ → 1
```

$\frac{18}{5}$

```
% // N
```

3.6

Next we find the probability density function that a random walker *first* reaches the lattice point {1, 1} after exactly *n* steps.

```
SeriesCoefficient[2 ξ^2/(9 - 4 ξ^2), {ξ, 0, n}] //
 FullSimplify[#, n ∈ Integers && n > 0 && Mod[n, 2] == 0] &
```

$2^{-1+n}\,3^{-n}$

We plot this density function in the next cell. Notice that the probabilities exist only for even values of *n* and that they quickly approach zero.

```
DiscretePlot[2^(-1+n) 3^(-n), {n, 2, 20, 2}, PlotRange → All, AxesOrigin → {0, 0}]
```



And, of course, this density function may be easily summed to obtain the 2/5 probability that the random walker will ever reach the lattice point {1, 1}.

```
Sum[2^(-1+n) 3^(-n), {n, 2, ∞, 2}]
```

$\frac{2}{5}$

Check Using Simulation

Using the following compiled function we can check our solution by performing a simulation experiment. Here *trials* is the number of trials we run, *steps* is the maximum number of steps over which we will track the random walker, and *target* is the lattice point we want to reach.

```
cfSim = Compile[{{trials, _Integer}, {steps, _Integer}, {target, _Integer, 1}},
   Module[{hit = 0, p = {0, 0}, cnt = 0, eval = 0},
    Do[cnt = 0;
     p = {{1, 0}, {0, 1}, {-1, 0}}[[RandomInteger[{1, 3}]]];
     While[cnt++ < steps && p ≠ target,
      p = p + {{1, 0}, {0, 1}, {-1, 0}}[[RandomInteger[{1, 3}]]]];
     If[p == target, hit++; eval = eval + cnt], {trials}];
    N[{hit / trials, eval / hit}]], RuntimeAttributes → Listable,
   Parallelization → True, CompilationTarget → "C"];
```

We run 100 million trials in the next cell using parallel kernels.

```
Mean@cfSim[Table[12 500 000, {8}], 100, {1, 1}]
```

```
{0.399893, 3.59997}
```

These values for the probability and the expected number of steps closely match those found in the analytic solution.

The Probability Of Reaching {3, 5}

To show how this method of solution may be generalized to any lattice point we investigate the probability of the random walker ever reaching the point {3, 5}.

Analytic Solution

Proceeding as above we first find the integral needed to obtain the generating function giving the probability that the random walker will be found at the point {3, 5} after a specified number of steps.

```
Assuming[0 < ξ < 1 && t > 0,
 HoldForm[Integrate[##, {t, 0, ∞}]] & @@ {1/(2 π)^2 Integrate[
    Exp[-(1 - ξ/3 (ⅇ^(ⅈ y) + 2 Cos[x])) t] Exp[-ⅈ (3 x + 5 y)], {x, -π, π}, {y, -π, π} ]}]
```

$$\int_0^\infty \frac{e^{-t}\, t^5\, \xi^5\, \text{BesselI}\left[3, \frac{2 t \xi}{3}\right]}{29160}\, dt$$

Evaluating this integral we obtain the generating function.

```
Assuming[ 0 < ξ < 1,
 Integrate[(ⅇ^-t t^5 ξ^5 BesselI[3, (2 t ξ)/3])/29160, {t, 0, ∞}] // FullSimplify]
```

$$\frac{84\, \xi^8 \left(18 + \xi^2\right)}{\left(9 - 4\, \xi^2\right)^{11/2}}$$

Dividing this by the corresponding generating function giving the probability that the random walker will be found at the origin after a specified number of steps, we obtain the generating function for the probability that the walker will *first* reach the lattice point {3, 5} after a specified number of steps.

```
(84 ξ^8 (18 + ξ^2) / (9 - 4 ξ^2)^(11/2)) / (3 / Sqrt[9 - 4 ξ^2]) // Simplify
```

$$\frac{28\,\xi^8\,(18+\xi^2)}{(9-4\,\xi^2)^5}$$

Evaluating this generating function at the point $\xi \to 1$ we obtain the probability that the random walker will ever reach the point {3, 5}.

```
28 ξ^8 (18 + ξ^2) / (9 - 4 ξ^2)^5 /. ξ → 1
```

$$\frac{532}{3125}$$

```
% // N
```

0.17024

Expanding this generating function as a Taylor's series about the point $\xi = 0$ we can read off the exact probabilities that the random walker *first* reaches the point {3, 5} after the indicated number of steps.

```
Series[28 ξ^8 (18 + ξ^2) / (9 - 4 ξ^2)^5, {ξ, 0, 18}]
```

$$\frac{56\,\xi^8}{6561}+\frac{1148\,\xi^{10}}{59\,049}+\frac{14\,000\,\xi^{12}}{531\,441}+\frac{132\,160\,\xi^{14}}{4\,782\,969}+\frac{1\,066\,240\,\xi^{16}}{43\,046\,721}+\frac{7\,727\,104\,\xi^{18}}{387\,420\,489}+O[\xi]^{19}$$

Taking the derivative of this generating function, evaluating it at $\xi \to 1$, and multiplying by the normalizing constant we get the expected number of steps required for the walker to *first* reach the point {3, 5}, given that he ever reaches this point.

```
3125/532 D[28 ξ^8 (18 + ξ^2) / (9 - 4 ξ^2)^5, ξ] /. ξ → 1
```

$$\frac{306}{19}$$

```
% // N
```

16.1053

Check Using Simulation

Performing a simulation experiment we run 100 million trials to check our analytic solution.

```
Mean@cfSim[Table[12 500 000, {8}], 100, {3, 5}]
```

{0.170282, 16.1055}

These values for the probability and the expected number of steps closely match those found in the analytic solution.

The Probability Of Reaching {12, 12}

We show another example by investigating the probability of the random walker ever reaching the point {12, 12}.

Analytic Solution

Proceeding as above we first find the integral needed to obtain the generating function giving the probability that the random walker will be found at the point {12, 12} after a specified number of steps.

```
Assuming[0 < ξ < 1 && t > 0,
 HoldForm[Integrate[##, {t, 0, ∞}]] & @@ FullSimplify[{1/(2 π)^2 Integrate[
      Exp[-(1 - ξ/3 (ⅇ^(ⅈ y) + 2 Cos[x])) t] Exp[-ⅈ (12 x + 12 y)], {x, -π, π}, {y, -π, π} ]}]]
```

$$\int_0^\infty \frac{1}{254\,561\,089\,305\,600} e^{-t}\, t\, \xi \left(t\, \xi \left(2\,357\,047\,123\,200 + t^2 \xi^2 \left(107\,138\,505\,600 + 1\,234\,517\,760\, t^2 \xi^2 + 4\,762\,800\, t^4 \xi^4 + 5670\, t^6 \xi^6 + t^8 \xi^8\right)\right) \mathrm{BesselI}\left[0, \frac{2 t \xi}{3}\right] - 54 \left(130\,947\,062\,400 + t^2 \xi^2 \left(13\,226\,976\,000 + 264\,539\,520\, t^2 \xi^2 + 1\,693\,440\, t^4 \xi^4 + 3675\, t^6 \xi^6 + 2\, t^8 \xi^8\right)\right) \mathrm{BesselI}\left[1, \frac{2 t \xi}{3}\right]\right) dt$$

Note that the complicated integrand in the prior cell can be expressed more simply as the right hand side of the following expression:

```
Block[{x = 12, y = 12},
 1/254 561 089 305 600 ⅇ^-t t ξ (t ξ (2 357 047 123 200 + t^2 ξ^2 (107 138 505 600 +
          1 234 517 760 t^2 ξ^2 + 4 762 800 t^4 ξ^4 + 5670 t^6 ξ^6 + t^8 ξ^8)) BesselI[0, 2 t ξ/3] -
      54 (130 947 062 400 + t^2 ξ^2 (13 226 976 000 + 264 539 520 t^2 ξ^2 + 1 693 440 t^4 ξ^4 +
          3675 t^6 ξ^6 + 2 t^8 ξ^8)) BesselI[1, 2 t ξ/3]) ==
   3^-y ⅇ^-t (t ξ)^y/y! BesselI[x, 2 t ξ/3] // FullSimplify[#, 0 < ξ < 1 && t > 0] &]
```

```
True
```

Evaluating this integral we obtain the generating function.

```
Assuming[ 0 < ξ < 1,
 ReleaseHold[%] // FullSimplify]
```

$$\frac{8\,112\,468\,\xi^{24}}{\left(9-4\,\xi^2\right)^{25/2}}$$

Dividing this by the corresponding generating function giving the probability that the random walker will be found at the origin after a specified number of steps, we obtain the generating function for the probability that the walker will *first* reach the lattice point {12, 12} after a specified number of steps.

$$\% \Big/ \left(\frac{3}{\sqrt{9-4\,\xi^2}}\right) \text{ // FullSimplify[\#, } 0<\xi<1\text{] \&}$$

$$\frac{2\,704\,156\,\xi^{24}}{\left(9-4\,\xi^2\right)^{12}}$$

Evaluating this generating function at the point $\xi \to 1$ we obtain the probability that the random walker will ever reach the point {12, 12}.

$$\frac{2\,704\,156\,\xi^{24}}{\left(9-4\,\xi^2\right)^{12}} \text{ /. } \xi \to 1 \text{ // Simplify}$$

$$\frac{2\,704\,156}{244\,140\,625}$$

```
% // N
```

0.0110762

Expanding this generating function as a Taylor's series about the point $\xi = 0$ we can read off the exact probabilities that the random walker *first* reaches the point {12, 12} after the indicated number of steps.

$$\text{Series}\left[\frac{2\,704\,156\,\xi^{24}}{\left(9-4\,\xi^2\right)^{12}}, \{\xi, 0, 30\}\right]$$

$$\frac{2\,704\,156\,\xi^{24}}{282\,429\,536\,481}+\frac{43\,266\,496\,\xi^{26}}{847\,288\,609\,443}+\frac{1\,124\,928\,896\,\xi^{28}}{7\,625\,597\,484\,987}+\frac{62\,996\,018\,176\,\xi^{30}}{205\,891\,132\,094\,649}+O[\xi]^{31}$$

Taking the derivative of this generating function, evaluating it at $\xi \to 1$, and multiplying by the normalizing constant we get the expected number of steps required for the walker to *first* reach the point {12, 12}, given that he ever reaches this point.

$$\frac{244\,140\,625}{2\,704\,156}\,D\left[\frac{2\,704\,156\,\xi^{24}}{\left(9-4\,\xi^2\right)^{12}}, \xi\right] \text{ /. } \xi \to 1$$

$$\frac{216}{5}$$

```
% // N
```

43.2

Check Using Simulation

Performing a simulation experiment we run 100 million trials to check our analytic solution.

```
Mean@cfSim[Table[12 500 000, {8}], 250, {12, 12}]
{0.0110754, 43.1897}
```

These values for the probability and the expected number of steps closely match those found in the analytic solution.

The Probability Of Reaching {7, 11}

We show another example by investigating the probability of the random walker ever reaching the point {7, 11}.

Analytic Solution

Proceeding as above we first find the integral needed to obtain the generating function giving the probability that the random walker will be found at the point {7, 11} after a specified number of steps.

```
Assuming[0 < ξ < 1 && t > 0,
 HoldForm[Integrate[##, {t, 0, ∞}]] & @@ FullSimplify[{1/(2 π)^2 Integrate[
     Exp[-(1 - ξ/3 (ⅇ^(ⅈ y) + 2 Cos[x])) t] Exp[-ⅈ (7 x + 11 y)], {x, -π, π}, {y, -π, π} ]}]]
```

$$\int_0^\infty \frac{1}{7\,071\,141\,369\,600} e^{-t} t^6 \xi^6 \left(t \xi \left(29\,160 + 540 t^2 \xi^2 + t^4 \xi^4\right) \text{BesselI}\left[1, \frac{2 t \xi}{3}\right] - 36 \left(4860 + 180 t^2 \xi^2 + t^4 \xi^4\right) \text{BesselI}\left[2, \frac{2 t \xi}{3}\right]\right) dt$$

Note that this complicated integrand can be expressed more simply as the right hand side of the following expression:

```
Block[{x = 7, y = 11},
 1/7 071 141 369 600 ⅇ^-t t^6 ξ^6 (t ξ (29 160 + 540 t^2 ξ^2 + t^4 ξ^4) BesselI[1, (2 t ξ)/3] -
     36 (4860 + 180 t^2 ξ^2 + t^4 ξ^4) BesselI[2, (2 t ξ)/3]) ==
   (3^-y ⅇ^-t (t ξ)^y)/y! BesselI[x, (2 t ξ)/3] // FullSimplify[#, 0 < ξ < 1 && t > 0] &]
True
```

Evaluating this integral we obtain the generating function.

```
Assuming[ 0 < ξ < 1,
 int0711 = ReleaseHold[%] // FullSimplify]
```

$$\frac{15\,912\,\xi^{18}\left(486+81\,\xi^{2}+\xi^{4}\right)}{\left(9-4\,\xi^{2}\right)^{23/2}}$$

Dividing this by the corresponding generating function giving the probability that the random walker will be found at the origin after a specified number of steps, we obtain the generating function for the probability that the walker will *first* reach the lattice point {7, 11} after a specified number of steps.

$$\text{int0711}\Big/\left(\frac{3}{\sqrt{9-4\,\xi^{2}}}\right)\text{ // FullSimplify[\#, 0 < }\xi\text{ < 1] \&}$$

$$\frac{5304\,\xi^{18}\left(486+81\,\xi^{2}+\xi^{4}\right)}{\left(9-4\,\xi^{2}\right)^{11}}$$

Evaluating this generating function at the point $\xi \to 1$ we obtain the probability that the random walker will ever reach the point {7, 11}.

$$\frac{5304\,\xi^{18}\left(486+81\,\xi^{2}+\xi^{4}\right)}{\left(9-4\,\xi^{2}\right)^{11}}\text{ /. }\xi\to 1\text{ // Simplify}$$

$$\frac{3\,012\,672}{48\,828\,125}$$

```
% // N
```

0.0616995

Expanding this generating function as a Taylor's series about the point $\xi = 0$ we can read off the exact probabilities that the random walker *first* reaches the point {7, 11} after the indicated number of steps.

$$\text{Series}\left[\frac{5304\,\xi^{18}\left(486+81\,\xi^{2}+\xi^{4}\right)}{\left(9-4\,\xi^{2}\right)^{11}},\ \{\xi, 0, 26\}\right]$$

$$\frac{3536\,\xi^{18}}{43\,046\,721}+\frac{160\,888\,\xi^{20}}{387\,420\,489}+\frac{11\,903\,944\,\xi^{22}}{10\,460\,353\,203}+\frac{211\,049\,696\,\xi^{24}}{94\,143\,178\,827}+\frac{1\,003\,827\,968\,\xi^{26}}{282\,429\,536\,481}+O[\xi]^{27}$$

Taking the derivative of this generating function, evaluating it at $\xi \to 1$, and multiplying by the normalizing constant we get the expected number of steps required for the walker to *first* reach the point {7, 11}, given that he ever reaches this point.

$$\frac{48\,828\,125}{3\,012\,672}\,D\left[\frac{5304\,\xi^{18}\left(486+81\,\xi^{2}+\xi^{4}\right)}{\left(9-4\,\xi^{2}\right)^{11}},\ \xi\right]\text{ /. }\xi\to 1$$

$$\frac{50\,967}{1420}$$

```
% // N
```

35.8923

Check Using Simulation

Performing a simulation experiment we run a billion trials to check our analytic solution.

```
Mean@cfSim[Table[125 000 000, {8}], 250, {7, 11}]
```

```
{0.0616933, 35.8919}
```

These values for the probability and the expected number of steps closely match those found in the analytic solution.

The Probability Of Reaching {11, 7}

We show one more example by investigating the probability of the random walker ever reaching the point {11, 7}.

Analytic Solution

Proceeding as above we first find the integral needed to obtain the generating function giving the probability that the random walker will be found at the point {11, 7} after a specified number of steps.

```
Assuming[0 < ξ < 1 && t > 0,
 HoldForm[Integrate[##, {t, 0, ∞}]] & @@ FullSimplify[{1/(2 π)^2 Integrate[
     Exp[-(1 - ξ/3 (ⅇ^(ⅈ y) + 2 Cos[x])) t] Exp[-ⅈ (11 x + 7 y)], {x, -π, π}, {y, -π, π} ]}]]
```

$$\int_0^{\infty} \frac{1}{33\,067\,440} e^{-t} \left(\left(11\,904\,278\,400 + 308\,629\,440\, t^2 \xi^2 + 2\,041\,200\, t^4 \xi^4 + 3780\, t^6 \xi^6 + t^8 \xi^8\right) \text{Hypergeometric0F1Regularized}\left[2, \frac{t^2 \xi^2}{9}\right] - 30 \left(793\,618\,560 + t^2 \xi^2 \left(35\,271\,936 + t^2 \xi^2 \left(504 + t^2 \xi^2\right)\left(756 + t^2 \xi^2\right)\right)\right) \text{Hypergeometric0F1Regularized}\left[3, \frac{t^2 \xi^2}{9}\right]\right) dt$$

Note that the complicated integrand in the prior cell can be expressed more simply as the right hand side of the following expression:

```
Block[{x = 11, y = 7},
  1/33067440 ⅇ^-t ((11904278400 + 308629440 t^2 ξ^2 + 2041200 t^4 ξ^4 + 3780 t^6 ξ^6 + t^8 ξ^8)
      Hypergeometric0F1Regularized[2, t^2 ξ^2/9] -
     30 (793618560 + t^2 ξ^2 (35271936 + t^2 ξ^2 (504 + t^2 ξ^2) (756 + t^2 ξ^2)))
      Hypergeometric0F1Regularized[3, t^2 ξ^2/9]) ==
   (3^-y ⅇ^-t (t ξ)^y)/y! BesselI[x, (2 t ξ)/3] // FullSimplify[#, 0 < ξ < 1 && t > 0] &]
```

True

Evaluating this integral we obtain the generating function.

```
Assuming[ 0 < ξ < 1,
 int1107 = ReleaseHold[%] // FullSimplify]
```

$$\frac{1}{\xi^4 \left(9-4\xi^2\right)^{15/2}} 36 \left(645700815 \left(-3+\sqrt{9-4\xi^2}\right)+\xi^2 \left(-14348907 \left(-459+143\sqrt{9-4\xi^2}\right)+2\xi^2 \left(12155\xi^{14}+ 54\xi^{10}\left(328185-41984\sqrt{9-4\xi^2}\right)+28697814\left(-170+49\sqrt{9-4\xi^2}\right)-1240029 \xi^2\left(-3315+872\sqrt{9-4\xi^2}\right)+6804\xi^8\left(-21879+3712\sqrt{9-4\xi^2}\right)- 32805\xi^6\left(-21879+4480\sqrt{9-4\xi^2}\right)+98415\xi^4\left(-21879+5152\sqrt{9-4\xi^2}\right)+ 9\xi^{12}\left(-109395+8192\sqrt{9-4\xi^2}\right)\right)\right)\right)$$

Dividing this by the corresponding generating function giving the probability that the random walker will be found at the origin after a specified number of steps, we obtain the generating function for the probability that the walker will *first* reach the lattice point {11, 7} after a specified number of steps.

```
g1107 = int1107 / (3 / Sqrt[9 - 4 ξ^2]) // FullSimplify[#, 0 < ξ < 1] &
```

$$\frac{1}{\xi^4 \left(9-4\xi^2\right)^7} 12 \Big(645\,700\,815 \left(-3+\sqrt{9-4\xi^2}\right)+\xi^2 \Big(-14\,348\,907 \left(-459+143\sqrt{9-4\xi^2}\right)+2\xi^2 \Big(12\,155\,\xi^{14}+54\,\xi^{10}\left(328\,185-41\,984\sqrt{9-4\xi^2}\right)+28\,697\,814\left(-170+49\sqrt{9-4\xi^2}\right)-1\,240\,029\,\xi^2\left(-3315+872\sqrt{9-4\xi^2}\right)+6804\,\xi^8\left(-21\,879+3712\sqrt{9-4\xi^2}\right)-32\,805\,\xi^6\left(-21\,879+4480\sqrt{9-4\xi^2}\right)+98\,415\,\xi^4\left(-21\,879+5152\sqrt{9-4\xi^2}\right)+9\,\xi^{12}\left(-109\,395+8192\sqrt{9-4\xi^2}\right)\Big)\Big)\Big)$$

Evaluating this generating function at the point $\xi \to 1$ we obtain the probability that the random walker will ever reach the point {11, 7}. We note that this probability is irrational.

```
Limit[g1107, ξ → 1] // Simplify
```

$$-\frac{264\,135\,048}{78\,125}+1512\sqrt{5}$$

```
% // N
```

0.00616758

Expanding this generating function as a Taylor's series about the point $\xi = 0$ we can read off the exact probabilities that the random walker *first* reaches the point {11, 7} after the indicated number of steps.

```
Series[g1107, {ξ, 0, 26}]
```

$$\frac{3536\,\xi^{18}}{43\,046\,721}+\frac{314\,704\,\xi^{20}}{1\,162\,261\,467}+\frac{5\,275\,984\,\xi^{22}}{10\,460\,353\,203}+\frac{65\,797\,888\,\xi^{24}}{94\,143\,178\,827}+\frac{75\,509\,920\,\xi^{26}}{94\,143\,178\,827}+O[\xi]^{27}$$

Taking the derivative of this generating function, evaluating it at $\xi \to 1$, and multiplying by the normalizing constant we get the expected number of steps required for the walker to *first* reach the point {11, 7}, given that he ever reaches this point.

```
1 / Limit[g1107, ξ → 1] D[g1107, ξ] /. ξ → 1 // FullSimplify
```

$$\frac{9\left(1\,516\,630\,937+949\,609\,375\sqrt{5}\right)}{1\,104\,787\,490}$$

```
% // N
```

29.6529

Check Using Simulation

Performing a simulation experiment we run 100 million trials to check our analytic solution.

```
Mean@cfSim[Table[12 500 000, {8}], 250, {11, 7}]
```

{0.00615456, 29.665}

These values for the probability and the expected number of steps closely match those found in the analytic solution.

The Probability Of Reaching {*n*, 0}

If we stay on the *x*-axis we can easily find the general form of the probability generating function giving the chance that the random walker *first* arrives at the point {*n*, 0} after a specified number of steps.

In this case it is not too hard to see that the integrand is always $e^{-t} I_n\left(\frac{2 t \xi}{3}\right)$. In the next cell we carry out the calculations for $n = 1, 2 \ldots, 5$.

```
ParallelTable[FullSimplify[Sqrt[9 - 4 ξ^2]/3
    Integrate[ⅇ^-t BesselI[n, (2 t ξ)/3], {t, 0, ∞}, Assumptions → 0 < ξ < 1]], {n, 1, 5}]
```

$$\left\{-\frac{-3+\sqrt{9-4\xi^2}}{2\xi}, \frac{9-2\xi^2-3\sqrt{9-4\xi^2}}{2\xi^2}, \frac{\xi^2\left(-9+\sqrt{9-4\xi^2}\right)-9\left(-3+\sqrt{9-4\xi^2}\right)}{2\xi^3}, \frac{2\xi^4+6\xi^2\left(-6+\sqrt{9-4\xi^2}\right)-27\left(-3+\sqrt{9-4\xi^2}\right)}{2\xi^4}, -\frac{1}{2\xi^5}\left(\xi^4\left(-15+\sqrt{9-4\xi^2}\right)-27\xi^2\left(-5+\sqrt{9-4\xi^2}\right)+81\left(-3+\sqrt{9-4\xi^2}\right)\right)\right\}$$

Evaluating these at $\xi \to 1$ we obtain:

```
% /. ξ → 1 // FullSimplify
```

$$\left\{\frac{1}{2}\left(3-\sqrt{5}\right), \frac{1}{2}\left(7-3\sqrt{5}\right), 9-4\sqrt{5}, \frac{1}{2}\left(47-21\sqrt{5}\right), \frac{1}{2}\left(123-55\sqrt{5}\right)\right\}$$

But this is equal to the right hand side of the following expression.

```
% == Table[1/GoldenRatio^(2 n), {n, 1, 5}] // FullSimplify
```

True

So we see that the probability that a random walker *first* arrives at the point {*n*, 0} is given by $1/\phi^{2n}$, where ϕ is the golden ratio.

Check Using Simulation

Performing a simulation experiment we run 100 million trials to check our analytic solution.

```
Mean@cfSim[Table[12 500 000, {8}], 250, {4, 0}]
```

{0.0212843, 5.36673}

$$\frac{1}{2}\left(47-21\sqrt{5}\right) \text{ // N}$$

0.0212862

```
Mean@cfSim[Table[12 500 000, {8}], 250, {5, 0}]
```

{0.0081267, 6.7076}

$$\frac{1}{2}\left(123-55\sqrt{5}\right) \text{ // N}$$

0.00813062

The values of the probabilities closely match those found in the analytic solutions.

The Probability Of Reaching {*n*, 1}

We can easily find the general form of the probability generating function giving the chance that the random walker *first* arrives at the point {*n*, 1} after a specified number of steps.

In this case it is not too hard to see that the integrand is always $\frac{1}{3}\xi e^{-t} t I_n\left(\frac{2 t \xi}{3}\right)$. In the next cell we carry out the calculations for $n = 1, 2 ..., 8$.

```
ParallelTable[FullSimplify[Sqrt[9 - 4 ξ^2]/3
   Integrate[1/3 ξ ⅇ^-t t BesselI[n, (2 t ξ)/3], {t, 0, ∞}, Assumptions → 0 < ξ < 1]], {n, 1, 8}]
```

$$\left\{\frac{2\,\xi^2}{9-4\,\xi^2},\ \frac{-9\left(-3+\sqrt{9-4\,\xi^2}\right)+2\,\xi^2\left(-9+2\sqrt{9-4\,\xi^2}\right)}{-18\,\xi+8\,\xi^3},\right.$$

$$\frac{3}{4}\left(2+\frac{1}{3-2\,\xi}+\frac{1}{3+2\,\xi}+\frac{4\left(-3+\sqrt{9-4\,\xi^2}\right)}{\xi^2}\right),\ \frac{1}{2\,\xi^3\left(9-4\,\xi^2\right)}$$

$$\left(18\,\xi^2\left(30-7\sqrt{9-4\,\xi^2}\right)+243\left(-3+\sqrt{9-4\,\xi^2}\right)+\xi^4\left(-90+8\sqrt{9-4\,\xi^2}\right)\right),$$

$$-\frac{2\left(-9+5\,\xi^2\right)}{-9+4\,\xi^2}-\frac{9\left(-7+\sqrt{9-4\,\xi^2}\right)}{\xi^2}+\frac{54\left(-3+\sqrt{9-4\,\xi^2}\right)}{\xi^4},$$

$$\frac{1}{2\,\xi^5\left(-9+4\,\xi^2\right)}\,3\left(\xi^4\left(945-153\sqrt{9-4\,\xi^2}\right)-\right.$$

$$\left.1215\left(-3+\sqrt{9-4\,\xi^2}\right)+\xi^6\left(-70+4\sqrt{9-4\,\xi^2}\right)+54\,\xi^2\left(-63+16\sqrt{9-4\,\xi^2}\right)\right),$$

$$\frac{36-14\,\xi^2}{9-4\,\xi^2}+\frac{54\left(24-5\sqrt{9-4\,\xi^2}\right)}{\xi^4}+\frac{18\left(-10+\sqrt{9-4\,\xi^2}\right)}{\xi^2}+\frac{729\left(-3+\sqrt{9-4\,\xi^2}\right)}{\xi^6},$$

$$\frac{1}{2\,\xi^7\left(9-4\,\xi^2\right)}\left(36\,\xi^6\left(252-31\sqrt{9-4\,\xi^2}\right)+1458\,\xi^2\left(108-29\sqrt{9-4\,\xi^2}\right)+\right.$$

$$\left.\left.45\,927\left(-3+\sqrt{9-4\,\xi^2}\right)+2\,\xi^8\left(-189+8\sqrt{9-4\,\xi^2}\right)+486\,\xi^4\left(-126+25\sqrt{9-4\,\xi^2}\right)\right)\right\}$$

Evaluating these at $\xi\to 1$ we obtain:

```
% /. ξ → 1 // FullSimplify
```

$$\left\{\frac{2}{5},\ \frac{1}{10}\left(-9+5\sqrt{5}\right),\ -\frac{33}{5}+3\sqrt{5},\ -\frac{279}{10}+\frac{25\sqrt{5}}{2},\right.$$

$$\left.-\frac{503}{5}+45\sqrt{5},\ -\frac{1677}{5}+150\sqrt{5},\ -\frac{5333}{5}+477\sqrt{5},\ -\frac{32\,859}{10}+\frac{2939\sqrt{5}}{2}\right\}$$

The general form of the solution being as follows:

```
FindSequenceFunction[%, n] // FullSimplify[#, n ∈ Integers && n > 0] &
```

$$\frac{1}{5}\left(\frac{2}{3+\sqrt{5}}\right)^{n}\left(3+\sqrt{5}\,n\right)$$

So we see that the probability that a random walker *first* arrives at the point {*n*, 1} is given by:

$$\frac{1}{5}\left(\frac{2}{3+\sqrt{5}}\right)^n\left(\sqrt{5}\, n+3\right)$$

Check Using Simulation

Performing a simulation experiment we run 100 million trials to check our analytic solution.

```
Mean@cfSim[Table[12 500 000, {8}], 250, {4, 1}]
```

```
{0.0508281, 7.36656}
```

$$\frac{1}{5}\left(\frac{2}{3+\sqrt{5}}\right)^n\left(3+\sqrt{5}\, n\right) \text{ /. } n \to 4 \text{ // N}$$

```
0.0508497
```

```
Mean@cfSim[Table[12 500 000, {8}], 250, {7, 1}]
```

```
{0.00442714, 11.3225}
```

$$\frac{1}{5}\left(\frac{2}{3+\sqrt{5}}\right)^n\left(3+\sqrt{5}\, n\right) \text{ /. } n \to 7 \text{ // N}$$

```
0.00442527
```

The values of the probabilities closely match those found in the analytic solutions.

The Probability Of Reaching {*n*, *n*}

If we stay on the diagonal so that $x = y$ the integrations are straight forward and we can easily find the general form of the probability generating function giving the chance that the random walker *first* arrives at the point $\{n, n\}$ after a specified number of steps.

In the next cell we carry out the calculations for $n = 1, 2 \ldots, 6$.

$$\text{tab = ParallelTable}\Big[\text{Assuming}\Big[0 < \xi < 1 \ \&\& \ t > 0, \text{FullSimplify}\Big[\text{ReleaseHold@HoldForm[Integrate[\#\#, \{t, 0, }\infty\}]] \ \& @@ \text{FullSimplify}\Big[\Big\{\frac{1}{(2\pi)^2}\text{Integrate}\Big[\text{Exp}\Big[-\Big(1-\frac{\xi}{3}\left(e^{i y}+2\,\text{Cos}[x]\right)\Big)t\Big]\, \text{Exp}[-i\,(n\,x+n\,y)], \{x, -\pi, \pi\}, \{y, -\pi, \pi\}\Big]\Big\}\Big] \Big/ \frac{3}{\sqrt{9-4\,\xi^2}}\Big]\Big], \{n, 1, 6\}\Big]$$

$$\left\{\frac{2\,\xi^2}{9-4\,\xi^2}, \frac{6\,\xi^4}{\left(9-4\,\xi^2\right)^2}, \frac{20\,\xi^6}{\left(9-4\,\xi^2\right)^3}, \frac{70\,\xi^8}{\left(9-4\,\xi^2\right)^4}, \frac{252\,\xi^{10}}{\left(9-4\,\xi^2\right)^5}, \frac{924\,\xi^{12}}{\left(9-4\,\xi^2\right)^6}\right\}$$

It is easy to deduce the form of the coefficient in the numerator.

```
FindSequenceFunction[{2, 6, 20, 70, 252, 924}, n]
```

```
Binomial[2 n, n]
```

So the general formula for this probability generating function is:

$$P[\{n, n\}] = \binom{2n}{n} \frac{\xi^{2n}}{(9 - 4\,\xi^2)^n}$$

This means that the probability that a random walker will ever reach the diagonal lattice point {*n*, *n*} is:

```
Limit[Binomial[2 n, n] ξ^(2 n)/(9 - 4 ξ^2)^n, ξ → 1]
```

```
5^-n Binomial[2 n, n]
```

So this probability is:

$$5^{-n} \binom{2n}{n}$$

We check this in the case of $n = 12$ and we see that it matches the result we obtained previously.

```
5^-n Binomial[2 n, n] /. n → 12
```

$$\frac{2\,704\,156}{244\,140\,625}$$

This is a very powerful result. Suppose we wanted to find the mode of the number of steps required (i.e., the most likely number of steps) for a random walker to *first* reaches the point {100, 100}. We compute this in the next two cells.

```
Block[{n = 100},
 mode = Table[SeriesCoefficient[Binomial[2 n, n] ξ^(2 n)/(9 - 4 ξ^2)^n, {ξ, 0, s}], {s, 200, 400}];]
```

```
Position[mode, Max[mode]] + 199
```

```
{{358}}
```

So the result is 358 steps.

The Probability Of Reaching {*x*, *y*}

With the knowledge gained in the prior sections we can now find a general function that will give us either 1) the probability generating function for the probability that a random walker *first* reaches the lattice position {*x*, *y*} after a specified number of steps, or 2) the probability that the random walker *ever reaches* the lattice position {*x*, *y*}.

We first find the integrand in terms of modified Bessel functions for the lattice Green function giving the probability that the random walker will be found at the lattice position {*x*, *y*} after a specified number of steps. To see the form of the integrand we examine the lattice positions {1, *n*} for $n = 1, 2 \ldots, 8$.

```
ParallelTable[1/(2 π)^2 Integrate[Exp[-(1 - ξ/3 (ⅇ^(ⅈ y) + 2 Cos[x])) t] Exp[-ⅈ (x + n y)],
   {x, -π, π}, {y, -π, π}, Assumptions → 0 < ξ < 1 && t > 0], {n, 1, 8}]
```

$$\left\{\frac{1}{3} e^{-t}\, t\, \xi\, \text{BesselI}\left[1, \frac{2 t \xi}{3}\right], \frac{1}{18} e^{-t} t^2 \xi^2\, \text{BesselI}\left[1, \frac{2 t \xi}{3}\right], \frac{1}{162} e^{-t} t^3 \xi^3\, \text{BesselI}\left[1, \frac{2 t \xi}{3}\right], \frac{e^{-t} t^4 \xi^4\, \text{BesselI}\left[1, \frac{2 t \xi}{3}\right]}{1944}, \frac{e^{-t} t^5 \xi^5\, \text{BesselI}\left[1, \frac{2 t \xi}{3}\right]}{29\,160}, \frac{e^{-t} t^6 \xi^6\, \text{BesselI}\left[1, \frac{2 t \xi}{3}\right]}{524\,880}, \frac{e^{-t} t^7 \xi^7\, \text{BesselI}\left[1, \frac{2 t \xi}{3}\right]}{11\,022\,480}, \frac{e^{-t} t^8 \xi^8\, \text{BesselI}\left[1, \frac{2 t \xi}{3}\right]}{264\,539\,520}\right\}$$

From the examples done in the prior sections (see above) we note that the argument of the modified Bessel function can always be expressed in the form $I_x\left(\frac{2 \xi t}{3}\right)$. The y value of the lattice position enters the Bessel function only through its coefficient. We extract these coefficients in the next cell.

```
% /. c_ _BesselI → c
```

$$\left\{\frac{1}{3} e^{-t}\, t\, \xi, \frac{1}{18} e^{-t} t^2 \xi^2, \frac{1}{162} e^{-t} t^3 \xi^3, \frac{e^{-t} t^4 \xi^4}{1944}, \frac{e^{-t} t^5 \xi^5}{29\,160}, \frac{e^{-t} t^6 \xi^6}{524\,880}, \frac{e^{-t} t^7 \xi^7}{11\,022\,480}, \frac{e^{-t} t^8 \xi^8}{264\,539\,520}\right\}$$

Now these coefficients can be expressed in a simple form as function of the y lattice position.

```
FindSequenceFunction[%, y] // FullSimplify[#, y ∈ Integers && y ≥ 0] &
```

$$\frac{3^{-y} e^{-t} (t\, \xi)^y}{y\,!}$$

So the general expression that must be integrated to find the required probabilities is

$$\frac{\sqrt{9 - 4 \xi^2}}{3^{y+1}\, y!} \int_0^\infty e^{-t} (\xi\, t)^y\, I_x\left(\frac{2\, t\, \xi}{3}\right) dt$$

We carry out the integration when x and y are left in symbolic form in the next cell. Note that we make the replacement $x \to |x|$ so that the function works on either side of the x-axis.

```
Assuming[0 < ξ < 1 && (x | y) ∈ Integers && x ≥ 0 && y ≥ 0,
 FullSimplify[
  Sqrt[9 - 4 ξ^2]/(3^(y+1) y!) Integrate[ⅇ^-t (t ξ)^y BesselI[x, (2 t ξ)/3], {t, 0, ∞}]] /. x → Abs[x]]
```

$$\frac{1}{y\,!} 3^{-1-y-\text{Abs}[x]}\, \xi^{y+\text{Abs}[x]} \sqrt{9 - 4 \xi^2}\, (y + \text{Abs}[x])\,!\; \text{Hypergeometric2F1Regularized}\left[\frac{1}{2}(1 + y + \text{Abs}[x]), \frac{1}{2}(2 + y + \text{Abs}[x]), 1 + \text{Abs}[x], \frac{4 \xi^2}{9}\right]$$

The probability generating function is then given by the following expression in terms of the hypergeometric ${}_2F_1$ regularized function.

$$\frac{\sqrt{9-4\,\xi^2}}{3^{|x|+y+1}\,y!}\,(|x|+y)!\,\xi^{|x|+y}\,{}_2\tilde{F}_1\left(\frac{1}{2}(y+|x|+1),\ \frac{1}{2}(y+|x|+2);\ |x|+1;\ \frac{4\,\xi^2}{9}\right)$$

We define this as a *Mathematica* function.

```
Clear[𝒫];
𝒫[x_, y_, ξ_: 1] :=
  Simplify[ Sqrt[9 - 4 ξ^2] / (3^(1+y+Abs[x]) y !) ξ^(y+Abs[x]) (y + Abs[x]) ! Hypergeometric2F1Regularized[
     1/2 (1 + y + Abs[x]), 1/2 (2 + y + Abs[x]), 1 + Abs[x], (4 ξ^2)/9]];
```

When the optional third argument is not supplied the function returns the probability that a random walker ever reaches the lattice position {*x*, *y*}. We test the function on all of our prior examples in the next cell.

```
𝒫 @@@ {{1, 1}, {3, 5}, {12, 12}, {7, 11}, {11, 7}}
```

$$\left\{\frac{2}{5},\ \frac{532}{3125},\ \frac{2\,704\,156}{244\,140\,625},\ \frac{3\,012\,672}{48\,828\,125},\ -\frac{264\,135\,048}{78\,125}+1512\,\sqrt{5}\right\}$$

```
𝒫 @@@ {{-1, 1}, {-3, 5}, {-12, 12}, {-7, 11}, {-11, 7}}
```

$$\left\{\frac{2}{5},\ \frac{532}{3125},\ \frac{2\,704\,156}{244\,140\,625},\ \frac{3\,012\,672}{48\,828\,125},\ -\frac{264\,135\,048}{78\,125}+1512\,\sqrt{5}\right\}$$

All of these values match those found from first principles in prior sections.

The formula also correctly reproduces the probabilities we found for the {*n*, 0} case.

```
𝒫 @@@ Table[{n, 0}, {n, 1, 5}] == Table[1 / GoldenRatio^(2 n), {n, 1, 5}] // FullSimplify
```

```
True
```

The formula reproduces the probabilities we found for the {*n*, 1} case.

```
𝒫 @@@ Table[{n, 1}, {n, 1, 10}] ==
  Table[1/5 (2/(3 + Sqrt[5]))^n (3 + Sqrt[5] n), {n, 1, 10}] // FullSimplify
```

```
True
```

The formula also reproduces all of the diagonal entries at the lattice points {*n*, *n*}.

```
𝒫 @@@ Table[{n, n}, {n, 1, 10}] == Table[5^-n Binomial[2 n, n], {n, 1, 10}] // FullSimplify
```

```
True
```

Finally, we note that if the optional third argument is supplied we obtain the probability generating function giving the probability that the random walker *first* reaches the lattice point {*x*, *y*} after the a specified number of steps. For example, at the lattice point {1, 1} this function is:

```
𝒫[1, 1, ξ]
```

$$\frac{2\,\xi^2}{9-4\,\xi^2}$$

Expanding this as a Taylor's series about the point $\xi = 0$ we can read off the probabilities that the random walker *first* reaches the lattice point {1, 1} after a specified number of steps.

```
Series[𝒫[1, 1, ξ], {ξ, 0, 18}]
```

$$\frac{2\,\xi^2}{9}+\frac{8\,\xi^4}{81}+\frac{32\,\xi^6}{729}+\frac{128\,\xi^8}{6561}+\frac{512\,\xi^{10}}{59\,049}+\frac{2048\,\xi^{12}}{531\,441}+\frac{8192\,\xi^{14}}{4\,782\,969}+\frac{32\,768\,\xi^{16}}{43\,046\,721}+\frac{131\,072\,\xi^{18}}{387\,420\,489}+O[\xi]^{19}$$

This being the result first given in a prior section.

We plot the lattice points in green together with the probability contours in the next cell.

```
Show[ContourPlot[𝒫[x, y], {x, -5.1, 5.1}, {y, -0.1, 5.1},
   Contours → Table[c, {c, 0.025, 1.0, 0.025}], MaxRecursion → 4, PlotRange → All],
 Graphics[{PointSize[Medium], Green, Table[Point[{x, y}], {x, -5, 5}, {y, 0, 5}]}],
 Axes → True, AspectRatio → Automatic, Background → LightGray, ImageSize → Full]
```



A Proof

In order to reach the point {*x*, *y*} we can start the last step of a random walk to that position at any of the three positions {{*x* – 1, *y*}, {*x*, *y* – 1}, {*x* + 1, *y*}}. This gives rise to the linear recurrence:

$$\frac{1}{3}\left(\mathcal{P}(x-1,\,y)+\mathcal{P}(x+1,\,y)+\mathcal{P}(x,\,y-1)\right)=\mathcal{P}(x,\,y)$$

The first row of the lattice is easily resolved into a formula for {*n*, 0} involving $1/\phi^{2n}$, where ϕ is the golden ratio (see above). In light of this we can prove that the general formula found in the prior section satisfies this recurrence as follows:

```
FullSimplify[
 𝒫[x + 1, y] + 𝒫[x - 1, y] + 𝒫[x, y - 1] == 3 𝒫[x, y] /. Abs[x_] → x, x ≥ 0 && y ≥ 0]
True
```

This completes the proof of the general formula.

References

Hughes, Barry D, *Random Walks and Random Environments – Volume I*, Oxford Science Publications, Clarendon Press, 1995